Polar Coordinates

```
> restart:with(plots):with(Student):with(MultivariateCalculus):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
```

Polar Graph Paper.

We use the command coordplot from the plots package to view a portion the polar plane. The option **grid[5,25]** specifies that there will be 5 equally spaced displayed radii from 0 to 4 and 25 equally spaced angles (each of measure $\frac{\pi}{12}$ or 15 degrees) from 0 to 2π.

```
> coordplot(polar,[0..4,0..2*Pi],labelling=true,grid=[5,25],view=
  [-4..4,-4..4],scaling=constrained);
```



Let's draw some polar graphs.

```
> polarplot(3,scaling=constrained);
```



This is the circle of raius 3.

```
> polarplot([r,Pi/3,r=-4..4],scaling=constrained,coords=polar);
```



This is the line which makes an angle of $\frac{\pi}{3}$ with the positive x-axis, so the slope of this line is $\tan\left(\frac{\pi}{3}\right) = \sqrt{3}$. We now get more adventurous and graph the polar function $r = \cos(5\,t)$ with t=0..2π. But the period of the function $y = \cos(5\,t)$ is $\frac{2\,\pi}{5}$, so we will just advance by 2*Pi/5 for 5 different steps. We also plot $y = \cos(5\,t)$, where we see that the cosine starts at 1, goes to -1, and then back to 1 again.

```
> plot(cos(5*t),t=0..2*Pi/5);
  p1:=polarplot(cos(5*t),t=0..2*Pi/5,scaling=constrained,color=
  red,thickness=8):
  display(p1);
```


1
0.5
0
-0.5
-1
0.2
0.4
0.6
0.8
1
1.2
t



Se how this matches with the cosine graph. We take a second step.

```
> p2:=polarplot(cos(5*t),t=2*Pi/5..4*Pi/5,scaling=constrained,
  color=green,thickness=8):
  display(p1,p2);
```



Now a third step.

```
> p3:=polarplot(cos(5*t),t=4*Pi/5..Pi,scaling=constrained,color=
  blue,thickness=8):
  p4:=polarplot(cos(5*t),t=Pi..6*Pi/5,scaling=constrained,color=
  blue,thickness=4):
  display(p1,p2,p3,p4);
```



We see that $t = 0..\pi$ give the entire graph, and that we have begun retracing the graph. We do a fourth step.

```
> p5:=polarplot(cos(5*t),t=6*Pi/5..8*Pi/5,scaling=constrained,
  color=violet,thickness=4):
  display(p1,p2,p3,p4,p5);
```



Finally, a fifth step.

```
> p6:=polarplot(cos(5*t),t=8*Pi/5..2*Pi,scaling=constrained,color=
  orange,thickness=4):
  display(p1,p2,p3,p4,p5,p6);
```



We see that we have traced out the curve twice. Next we do a spiral, the function $r = t$ with t=0..8π.

```
> polarplot(t,t=0..8*Pi,scaling=constrained);
```



Changing an Equation from Rectangular to Polar Coordinates.

We wish to change the equation $(x-1)^2+y^2=1$ to an equation in polar coordinates. We enter the equation.

```
> eqn:=(x-1)^2+y^2=1;
```

$$eqn := (x-1)^2+y^2=1$$

We use the ChangeOfVariables command on the left hand side of the equation to get the equation in polar form

```
> peqn:= ChangeOfVariables(lhs(eqn), [cartesian[x,y], polar[r,
  theta]])=1;
```

$$peqn := (r\cos(\theta)-1)^2+r^2\sin(\theta)^2=1$$

We solve the equation for r in terms of θ.

```
> reqn:=solve(peqn,r);
```

$$reqn := 0, \frac{2\cos(\theta)}{\cos(\theta)^2+\sin(\theta)^2}$$

We simplify the second solution.

```
> r:=simplify(reqn[2]);
```

$$r := 2\cos(\theta)$$